Math 918 Computer Worksheet 5

The m2 file with the commands listed in this worksheet is *Worksheet5.m2*.

1. (The Eliahou-Kervaire resolution) Let $R = \mathbb{Q}[x_1, \ldots, x_4]$ and

$$I = \langle\ x_1x_2x_4^4,\ x_1x_2x_3x_4^2,\ x_1x_3^6,\ x_1x_2x_3^2,\ x_2^6,\ x_1x_2^2,\ x_1^2\ \rangle.$$

We'll be working with a Position over coefficient order on free R-modules, so start by defining R with this ordering

```
R=QQ[x_1..x_4,MonomialOrder=>{Position=>Up,GRevLex=>4}]
```

You can define I as an object of type **Ideal** by typing

```
I = ideal(x_1*x_2*x_4^4,x_1*x_2*x_3*x_4^2,x_1*x_3^6,x_1*x_2*x_3^2,x_2^6,x_1*x_2^2,x_1^2)
```

or of type **MonomialIdeal** by typing

```
monI = monomialIdeal(x_1*x_2*x_4^4,x_1*x_2*x_3*x_4^2,x_1*x_3^6,x_1*x_2*x_3^2,x_2^6,x_1*x_2^2,x_1^2)
```

You can get the initial ideal of a given **Ideal** (which will be of type **MonomialIdeal**) as

```
monomialIdeal I
```

It is important to realize that certain commands will take as input objects of type **MonomialIdeal** (not Ideal), so you will not be able to apply such a function to the I above.

(a) Let's check that *Macaulay2* recognizes the three ideals above are the same.

```
I == mon I
I == monomialIdeal I
mon I == monomialIdeal I
```

(b) Check (using the criterion discussed in class) that I is Borel-fixed.

(c) Check using *Macaulay 2* that I is Borel-fixed. Employ the command `isBorel`. Use `viewHelp isBorel` to see what type of data is valid as input for the command.

(d) Compute the minimal free resolution and betti numbers of R/I as follows:

```
r = res I
r.dd
betti r
```

Note: the command `res I` will compute the resolution of R/I NOT of I. You can convince yourselves of this by computing `res(R/I)` and comparing.

(e) Compute the initial module of the first syzygy module of I as follows:

```
G = gens gb ker (r.dd_1)
leads = for i to (numcols G -1) list leadTerm G_i
M = image matrix leads
```

(f) Decompose M into a direct sum $M = \oplus I_i e_i$ and find the minimal free resolution of M using the minimal free resolutions of the ideals I_i.

(g) Check your work in part (f) using *Macaulay 2* i.e. compute `s = res M`.

(h) Compute `s = res M` and compare `s.dd` and `t.dd`. In particular, what is the relationship between the total Betti numbers of the resolutions s and t ? Can you think of a theoretical explanation for this?

2. (Build your own gin) The function below computes the generic initial ideal of an ideal I with high probability (it also gives a warning if returning a result that is unreliable).

```
gin = I -> (
   R:= ring I;
   n := # gens R;
   g:= random(R^1,R^{n:-1});
   F := map(R,R,g);
   genericI := F I;-- or:  genericI := substitute(I,g);
   generic = monomialIdeal genericI;
   good := isBorel generic;
   if not good then stderr << "--warning: potential generic initial ideal is not
                                                            Borel-fixed" << endl;
   return generic
 );
```

(a) Let I be your favorite homogeneous ideal (in a reasonably small number of variables, otherwise the code above may stall). Compute `gin(I)` for various monomial orderings, that is define I as an ideal in a ring with various monomial orderings and call the function above.

The function above is a simplified version of the function `gin` in the package `GenericInitialIdeals`. Alternatively, you can use the function (with options) in that package as follows

```
loadPackage "GenericInitialIdeals"
gin(I, MonomialOrder => Lex)
gin(I, MonomialOrder => GRevLex)
```

(b) By running several examples, come up with conjectural answers to the following questions:

- What is the relationship between $\operatorname{pd} I$ and $\operatorname{pd} \operatorname{gin}_{\mathrm{GrRevLex}} I$? (pd=projective dimension) How about the similar question for regularity?
- What is the relationship between $\operatorname{pd} I$ and $\operatorname{pd} \operatorname{gin}_{\mathrm{Lex}} I$? How about the similar question for regularity?
- What is the relationship between the betti numbers of I and those of $\operatorname{gin}(I)$ with respect to some monomial order?

3. The LCM lattice and multigraded Betti numbers of monomial ideals

Let $R = \mathbb{Q}[x, y]$ and consider the monomial ideal $I = \langle x^2, xy, y^3\rangle$. We give an approach to computing the multigraded Betti numbers of I using a combinatorial object called the LCM-lattice of I. This lattice, denoted by L_I, has as its elements the least common multiples of subsets of the generators of I and is ordered by divisibility. Every pair of elements in the lattice has a join (least common multiple) and a meet (greatest common divisor).

(a) Begin by constructing L_I using the functionality of the package `Posets`.

```
R=QQ[x,y]
I= ideal (x^2,x*y,y^3)
loadPackage "Posets"
L= lcmLattice I
texPoset(L,SuppressLabels=> false)
```

If you paste the output of the last command into a LaTeX file, you will see the Hasse diagram (pictorial representation) of L_I rendered below (you need to use package *tikz*).



(b) Compute the multigraded Betti numbers of I. Use the entries in the maps of the free resolutions produced by *Macaulay2* to figure out the multigraded shifts.

```
 betti res I
(res I).dd
```

(c) Recover these multigraded Betti numbers from the following formula for non-zero Betti numbers given by Gasharov-Peeva-Welker:

$$\beta_{i,\alpha}(R/I) = \dim_k \widetilde{H}_{i-2}(O(1,\alpha)_{L_I}), \text{ for } \alpha \in L_I.$$

Here $O(1, \alpha)_{L_I}$ is the order complex of the open interval $(1, \alpha)_{L_I}$. This open interval consists of all non-unit monomials in L_I that strictly divide α and the order complex $O(P)$ of a poset P is the abstract simplicial complex whose vertices are the elements of P and whose faces are the chains in the poset.

To simplify your work you may use commands similar to the following ($HH\_i = \widetilde{H}_i$)

```
O = orderComplex openInterval(L, 1_S, x^2*y)
for i to 5 list prune HH_i O
```